
SAS PROGRAMS 

FOR THE TABULATION OF POST OFFICE BOX DATA



�POST OFFICE BOX DATA TABULATION PROGRAMS



This library reference consists of two SAS programs, BOXES.SAS and BOXSQF.SAS.  These programs retabulate Postal Service data files provided in library references H�216 and H-278 to expand delivery groups as proposed by OCA witness Callow.



I.	Program BOXES.SAS



This program reads two input files that were provided by the Postal Service in LR-H-278: BOXESSEP.PRN and DSFSEPT97.PRN. This program produces the following:

a.	Tabulations of the PO Box Survey data similar to Tables 1 and 2 of USPS-T-24, based on H-278 data,

b.	Expansion factors (similar to Table 3 of USPS-T-24) to adjust the POB Survey data to the September 1997 Delivery Statistics File (DSF) installed box estimates,

c.	Tabulations (using the expansion factors) of the POB Survey data similar to Table 4 of USPS-T-24, broken out by CAG group (A-D, E-G, and H-L) to reflect the OCA's expanded delivery groups,

d.	Tabulations of the total number of installed boxes by CAG and delivery group, and

e.	Tabulations of the number of post offices in the POB Survey by CAG and delivery group.





II.	Program BOXSQF.SAS



This program retabulates the H-216 data file RENT.DATA to produce cost per square foot estimates for the fee groups proposed by OCA witness Callow.

�****************************************************************;

*       filename: BOXES.SAS                                    ;

*                                                               ;

*       THIS PROGRAM PERFORMS THE FOLLOWING OPERATIONS:         ;

*       1. PRODUCES TABLES ANALOGOUS TO USPS-T-24 TABLES 1 AND  ;

*          2 BASED ON THE LR-H-278 DATA SET.                    ;

*       2. PRODUCES EXPANSION FACTORS ANALOGOUS TO THOSE  OF    ;

*       USPS-T-24 TABLE 3 FOR USE WITH THE LR-H-278 DATA SET.   ;

*       3. PRODUCES THE EQUIVALENT OF USPS-T-24 TABLE 4 BASED   ;

*       ON H-278 DATA AND THE EXPANSION FACTORS.                ;

*       4. PRODUCES TABLES SIMILAR TO USPS-T-24 TABLE 4, BROKEN ;

*       OUT BY THE CAG GROUPINGS DEFINED IN DATA A, BELOW.      ;

*                                                               ;

*       Only H-278 data for delivery groups A-E are used in this;

*       program.  104 observations contained a '0' as the       ;

*       delivery group code and were dropped from analysis.     ;

****************************************************************;



filename in1 'c:\boxes\h-278\boxessep.prn';

filename in2 'c:\boxes\h-278\dsfsept97.prn';

proc format; *formats for the delivery group codes;

        value $dgrp

        'A'='City-A'

        'B'='City-B'

        'C'='City-Other'

        'D'='Non-city'

        'E'='Nondel.'

        ;



data a;

        infile in1 firstobs=2;

        input

        @9      cag     $char1.

        @25     group   $char1.

        @45     boxsqft 6.

        @62     worksqft        6.

        @79     boxinst1        6.

        @96     boxinst2        6.

        @113    boxinst3        6.

        @130    boxinst4        6.

        @147    boxinst5        6.

        @164    boxrent1        6.

        @181    boxrent2        6.

        @199    boxrent3        6.

        @216    boxrent4        6.

        @233    boxrent5        6.;



        tboxinst=sum(of boxinst1-boxinst5);

        tboxrent=sum(of boxrent1-boxrent5);



        if 'A'<=group<='E';     *delete records with invalid group;

        if cag~='';             *delete records with missing CAG;



        *assign cag groups*****************;

        if 'A'<=cag<='D' then cgroup='A-D';

        if 'E'<=cag<='G' then cgroup='E-G';

        if 'H'<=cag<='L' then cgroup='H-L';



        *save formatted delivery group as DGROUP;

        dgroup=put(group, $dgrp.);



****************************************************************;

*       Produce the equivalent of USPS-T-24 Tables 1 and 2 based;

*       on LR-H-278 data files. This matches the output at page ;

*       16 of LR-H-278.                                         ;

proc means data=a nway noprint;

        class dgroup;

        var boxinst1-boxinst5 boxrent1-boxrent5 tboxinst tboxrent;

        output out=tab1_2 sum=;

proc print;

        Title 'Estimated Boxes Installed (LR-H-278 POB Survey Data)';

        title2 '(Anlogous to Table 1 of USPS-T-24)';

        var dgroup boxinst1-boxinst5 tboxinst;

        sum boxinst1-boxinst5 tboxinst;

proc print;

        Title 'Estimated Boxes In Use (LR-H-278 POB Survey Data)';

        title2 '(Anlogous to Table 2 of USPS-T-24)';

        var dgroup boxrent1-boxrent5 tboxrent;

        sum boxrent1-boxrent5 tboxrent;





*******************************************************************;

*compute expansion factors analagous to those of USPS-T-24,Table 3.;

*Merge the expansion factors back with the full boxes data set to  ;

*produce the fboxes data set.                                      ;

*******************************************************************;



****************************************************************;

*Read the number of boxes installed by CAG and Group from the   ;

*september 1997 DSF.                                            ;

data dsf;

        infile in2 firstobs=2;

        input

        @9 group $char1.

        @28 cag $char1.

        @54 numbox 5.;

        dgroup=put(group, $dgrp.);

        drop group;

****************************************************************;

* compute DSF  installed box totals by cag and group.   ;

proc means data=dsf nway noprint;

  class dgroup cag;

  var numbox;

  output out=sdsf sum=;

proc sort data=sdsf; by dgroup cag;



****************************************************************;

* compute PO box study installed box totals by cag and group.   ;

proc means data=a noprint nway;

        class dgroup cag;

        id cgroup;

        var tboxinst;

        output out=sinst sum=tinst;

proc sort data=sinst; by dgroup cag;



****************************************************************;

* merge the PO box totals and the DSF installed box totals      ;

* by group and cag                                              ;

data efactor; merge sinst sdsf; by dgroup cag;

        *only compute expansion factors for cag or group        ;

        *combinations with installed boxes.;

        if tinst~=.;

        if tinst>0;

proc means data=efactor nway noprint;

        class dgroup cgroup;

        var numbox tinst;

        output out=efact2 sum=;



****************************************************************;

* compute the expansion factors                                 ;

data efact2; set efact2;

        fact=numbox/tinst;

proc print data=efact2;

        title 'Expansion Factors For Use With H-278 data';

        title2 '(Analogous to Table 3 of USPS-T-24, by CAG group)';

        var dgroup cgroup numbox tinst fact;



****************************************************************;

* add the expansion factor to each record in the POB data set   ;

proc sort data=efact2; by dgroup cgroup;

proc sort data=a; by dgroup cgroup;

data fboxes; merge efact2 a; by dgroup cgroup;

        drop _type_ _freq_ tinst numbox group;





*********************************************************************;

*sum the fboxes dataset to produce a table analogous to USPS-T-24,   ;

*Table 4.                                                            ;

*********************************************************************;

proc means data=fboxes noprint nway;

        class dgroup;

        weight fact;

        var boxrent1-boxrent5 tboxrent;

        output out=table4 sum=;

proc print data=table4;

        title 'Estimated Boxes In Use, Pre-MC96-3, based on H-278 data';

        title2 '(Analogous to Table 4 of USPS-T-24)';

        sum _freq_ boxrent1-boxrent5 tboxrent;



*********************************************************************;

*produce tables analogous to USPS-T-24 Table 4, broken out by CAG group;

*********************************************************************;

proc means data=fboxes noprint nway;

        class dgroup cgroup;

        weight fact;

        var boxrent1-boxrent5 tboxrent;

        output out=T4xCAG sum=;

proc print data=T4xCAG noobs;

        by dgroup;

        title 'Est. Pre-MC96-3 Boxes In Use By CAG Group, H-278 data';

        title2 '(Analogous to Table 4 of USPS-T-24, by CAG groupings)';

        var cgroup _freq_ boxrent1-boxrent5 tboxrent;

        sum _freq_ boxrent1-boxrent5 tboxrent;



*********************************************************************;

*produce table of total boxes in use by CAG and Group.               ;

*********************************************************************;

proc means data=fboxes noprint nway;

        class dgroup cag;

        weight fact;

        var boxrent1-boxrent5 tboxrent;

        output out=T4xCAG sum=;



proc freq data=t4xcag;

        title 'Est. Pre-MC96-3 Total Boxes In Use by CAG and Delivery Group';

        title2 'based on LR-H-278 data';

        weight tboxrent;

        tables cag*dgroup / nocol norow nopercent;



*********************************************************************;

*produce table of POB survey offices by cag and group                ;

*********************************************************************;

proc freq data=fboxes;

tables cag*dgroup /nopercent nocol norow;

title 'Offices in POB Survey by CAG and Delivery Group';

title2 'based on LR-H-278 data';

run;

�(Anlogous to Table 1 of USPS-T-24)

                                                                15:19 Thursday, October 16, 1997



   OBS    DGROUP        BOXINST1    BOXINST2    BOXINST3    BOXINST4    BOXINST5    TBOXINST



    1     City-A           35535        1987        1162        118          51        38853

    2     City-B           58079       16525        5899       1154         747        82404

    3     City-Other     4201907     2028034      718205     170547       40696      7159389

    4     Non-city       3577136     1548797      411307      35300        6680      5579220

    5     Nondel.         918567      323245       80017       5590        3680      1331099

                        ========    ========    ========    ========    ========    ========

                         8791224     3918588     1216590     212709       51854     14190965

�

                       Estimated Boxes In Use (LR-H-278 POB Survey Data)                      10

                               (Anlogous to Table 2 of USPS-T-24)

                                                                15:19 Thursday, October 16, 1997



   OBS    DGROUP        BOXRENT1    BOXRENT2    BOXRENT3    BOXRENT4    BOXRENT5    TBOXRENT



    1     City-A           26350        1644        922          96          28        29040

    2     City-B           49829       11966       4309         674         678        67456

    3     City-Other     3498063     1483084     491133      104946       21979      5599205

    4     Non-city       2928396     1217569     318872       25503        2829      4493169

    5     Nondel.         742423      244690      58774        3907         414      1050208

                        ========    ========    ========    ========    ========    ========

                         7245061     2958953     874010      135126       25928     11239078

�

                           Expansion Factors For Use With H-278 data                          11

                       (Analogous to Table 3 of USPS-T-24, by CAG group)

                                                                15:19 Thursday, October 16, 1997



                  OBS    DGROUP        CGROUP     NUMBOX      TINST      FACT



                    1    City-A         A-D       104384      38853    2.68664

                    2    City-B         A-D       202719      82404    2.46006

                    3    City-Other     A-D      6608169    3989487    1.65640

                    4    City-Other     E-G      3828343    2890257    1.32457

                    5    City-Other     H-L       333537     279645    1.19272

                    6    Non-city       A-D        70428      44937    1.56726

                    7    Non-city       E-G      2202548    1544979    1.42562

                    8    Non-city       H-L      4797066    3989304    1.20248

                    9    Nondel.        A-D        15983       6633    2.40962

                   10    Nondel.        E-G       336871     231101    1.45768

                   11    Nondel.        H-L      1357016    1093365    1.24114

�

                    Estimated Boxes In Use, Pre-MC96-3, based on H-278 data                   12

                              (Analogous to Table 4 of USPS-T-24)

                                                                15:19 Thursday, October 16, 1997



 OBS DGROUP     _TYPE_ _FREQ_   BOXRENT1   BOXRENT2   BOXRENT3   BOXRENT4 BOXRENT5   TBOXRENT



  1  City-A        1       29    70792.95    4416.84    2477.08    257.92    75.23    78020.01

  2  City-B        1      109   122582.46   29437.11   10600.41   1658.08  1667.92   165945.98

  3  City-Other    1     8080  5252862.88 2249832.56  747219.32 161947.22 34646.03  8446508.01

  4  Non-city      1    13354  3711735.07 1555055.48  408233.47  33524.73  3729.13  5712277.87

  5  Nondel.       1     3912   954373.46  315249.03   76101.68   5269.42   568.77  1351562.36

                       ====== =========== ========== ========== ========= ======== ===========

                        25484 10112346.82 4153991.01 1244631.96 202657.37 40687.08 15754314.23

�

                     Est. Pre-MC96-3 Boxes In Use By CAG Group, H-278 data                    13

                     (Analogous to Table 4 of USPS-T-24, by CAG groupings)

                                                                15:19 Thursday, October 16, 1997



---------------------------------------- DGROUP=City-A -----------------------------------------



    CGROUP    _FREQ_    BOXRENT1    BOXRENT2    BOXRENT3    BOXRENT4    BOXRENT5    TBOXRENT



     A-D        29      70792.95     4416.84     2477.08     257.917     75.2259    78020.01





---------------------------------------- DGROUP=City-B -----------------------------------------



   CGROUP    _FREQ_     BOXRENT1    BOXRENT2    BOXRENT3    BOXRENT4    BOXRENT5     TBOXRENT



    A-D        109     122582.46    29437.11    10600.41     1658.08     1667.92    165945.98





-------------------------------------- DGROUP=City-Other ---------------------------------------



CGROUP    _FREQ_     BOXRENT1      BOXRENT2      BOXRENT3     BOXRENT4    BOXRENT5     TBOXRENT



A-D        3541     3174929.53    1450575.26    491011.99    115387.84    27911.92    5259816.54

E-G        3730     1927796.67     752140.25    240966.80     44961.15     6206.93    2972071.80

H-L         809      150136.68      47117.04     15240.52      1598.24      527.18     214619.66

          ------    ----------    ----------    ---------    ---------    --------    ----------

DGROUP     8080     5252862.88    2249832.56    747219.32    161947.22    34646.03    8446508.01





--------------------------------------- DGROUP=Non-city ----------------------------------------



CGROUP    _FREQ_     BOXRENT1      BOXRENT2      BOXRENT3    BOXRENT4    BOXRENT5     TBOXRENT



A-D           36      33345.04      20433.95      7029.16     1065.74      155.16      62029.05

E-G         1449    1166834.55     550701.51    147967.62    16674.01     1860.43    1884038.12

H-L        11869    2511555.48     983920.02    253236.68    15784.98     1713.54    3766210.70

          ------    ----------    ----------    ---------    --------    --------    ----------

DGROUP     13354    3711735.07    1555055.48    408233.47    33524.73     3729.13    5712277.87





---------------------------------------- DGROUP=Nondel. ----------------------------------------



  CGROUP   _FREQ_      BOXRENT1    BOXRENT2     BOXRENT3     BOXRENT4   BOXRENT5     TBOXRENT



  A-D           7       6361.39      4595.14      1522.88      163.85      53.01      12696.28

  E-G         198     200871.04     62785.14     16267.69     2294.39     196.79     282415.04

  H-L        3707     747141.03    247868.76     58311.11     2811.18     318.97    1056451.04

           ------   -----------   ----------   ----------   ---------   --------   -----------

  DGROUP     3912     954373.46    315249.03     76101.68     5269.42     568.77    1351562.36

           ======   ===========   ==========   ==========   =========   ========   ===========

            25484   10112346.82   4153991.01   1244631.96   202657.37   40687.08   15754314.23

�

                  Est. Pre-MC96-3 Total Boxes In Use by CAG and Delivery Group                14

                                     based on LR-H-278 data     15:19 Thursday, October 16, 1997



                                     TABLE OF CAG BY DGROUP



                 CAG       DGROUP



                 Frequency‚City-A  ‚City-B  ‚City-Oth‚Non-city‚Nondel. ‚  Total

                          ‚        ‚        ‚er      ‚        ‚        ‚

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 A        ‚  78020 ‚  89940 ‚1379279 ‚      0 ‚      0 ‚1547239

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 B        ‚      0 ‚  53300 ‚ 943344 ‚   8098 ‚ 942.16 ‚1005684

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 C        ‚      0 ‚  19986 ‚1913124 ‚  24821 ‚  10101 ‚1968031

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 D        ‚      0 ‚ 2720.8 ‚1024070 ‚  29110 ‚   1653 ‚1057554

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 E        ‚      0 ‚      0 ‚1396990 ‚ 204416 ‚  29473 ‚1630880

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 F        ‚      0 ‚      0 ‚ 973211 ‚ 468080 ‚  57424 ‚1498714

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 G        ‚      0 ‚      0 ‚ 601871 ‚1211542 ‚ 195518 ‚2008931

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 H        ‚      0 ‚      0 ‚ 153636 ‚1483834 ‚ 212526 ‚1849996

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 J        ‚      0 ‚      0 ‚  32919 ‚1287203 ‚ 320841 ‚1640963

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 K        ‚      0 ‚      0 ‚  26421 ‚ 963414 ‚ 487619 ‚1477454

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 L        ‚      0 ‚      0 ‚ 1643.6 ‚  31760 ‚  35464 ‚  68868

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 Total       78020   165946  8446508  5712278  1351562  1.575E7







�

                        Offices in POB Survey by CAG and Delivery Group                       15

                                     based on LR-H-278 data     15:19 Thursday, October 16, 1997



                                     TABLE OF CAG BY DGROUP



                 CAG       DGROUP



                 Frequency‚City-A  ‚City-B  ‚City-Oth‚Non-city‚Nondel. ‚  Total

                          ‚        ‚        ‚er      ‚        ‚        ‚

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 A        ‚     29 ‚     56 ‚   1053 ‚      0 ‚      0 ‚   1138

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 B        ‚      0 ‚     42 ‚    637 ‚      5 ‚      1 ‚    685

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 C        ‚      0 ‚      9 ‚   1213 ‚     14 ‚      5 ‚   1241

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 D        ‚      0 ‚      2 ‚    638 ‚     17 ‚      1 ‚    658

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 E        ‚      0 ‚      0 ‚   1228 ‚    106 ‚     12 ‚   1346

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 F        ‚      0 ‚      0 ‚   1264 ‚    294 ‚     37 ‚   1595

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 G        ‚      0 ‚      0 ‚   1238 ‚   1049 ‚    149 ‚   2436

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 H        ‚      0 ‚      0 ‚    520 ‚   2186 ‚    314 ‚   3020

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 J        ‚      0 ‚      0 ‚    130 ‚   3284 ‚    712 ‚   4126

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 K        ‚      0 ‚      0 ‚    136 ‚   5759 ‚   2219 ‚   8114

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 L        ‚      0 ‚      0 ‚     23 ‚    640 ‚    462 ‚   1125

                 ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                 Total          29      109     8080    13354     3912    25484







�****************************************************************;

*       Program:  BOXSQF.SAS                                    ;

*                                                               ;

*       This program retabulates the H-216 data file RENT.DATA  ;

*       to produce cost per square foot estimates for the       ;

*       Delivery groups proposed by the OCA.                    ;

****************************************************************;



filename in1 't:\r97-1\libref\h-216\rent.data';

filename in1 'c:\trash\rent.data';



proc format; *formats for the delivery group codes;

        value $dgrp

        'A'='City-A'

        'B'='City-B'

        'C'='City-Other'

        'D'='Non-city'

        'E'='Nondel.'

        ;

data a;

  infile in1 dlm=',' firstobs=2;

  input group $ cag $ rcsf ra sf;

  if cag~='';   **eliminate records with missing cag or group;

  if group~=''; **eliminate records with missing cag or group;

        if group='A' then newgrp='A  ';

        if group='B' then newgrp='B  ';

        if group='E' then newgrp='E  ';

        if group='C' or group='D' then do;

                if 'A'<=cag<='D' then newgrp='CD1';

                if 'E'<=cag<='G' then newgrp='CD2';

                if 'H'<=cag<='L' then newgrp='CD3';

                end;

        *save formatted delivery group as DGROUP;

        dgroup=put(group, $dgrp.);

proc means data=a nway;

  class newgrp;

  title 'Rental Cost per SF, by NEWGRP, H-216 data';

  var rcsf;



*compare cag rcsf figures under old and new groupings;

data b; set a;

        if group='C' or group='D';

proc means data=b nway;

  class newgrp cag;

  title 'Rental Cost per SF, by NEWGRP & CAG, H-216 data';

  var rcsf;

proc means data=b nway;

        class dgroup cag;

        title 'Rental Cost per SF, by Delivery group & CAG, H-216 data';

        var rcsf;

proc means data=b nway;

        class dgroup;

        title 'Rental Cost per SF, by Delivery group, H-216 data';

        var rcsf;



run;





�                           Rental Cost per SF, by NEWGRP, H-216 data                           7

                                                                  10:21 Monday, October 20, 1997



          Analysis Variable : RCSF





          NEWGRP  N Obs      N          Mean       Std Dev       Minimum       Maximum

          ----------------------------------------------------------------------------

          A          30     30    23.4904980    17.1993379     0.0019685    64.0482433



          B         153    153    16.7430583    10.6920571     0.0051282    43.5236769



          CD1      3048   3048     9.0494744     6.9274203     0.0076923    35.7997936



          CD2      3782   3782     7.0475485     4.5347886     0.0076923    34.4827586



          CD3     13390  13390     5.7872161     2.7621283     0.8640000    26.6166667



          E        4170   4170     7.1935801     3.8123217     1.0666667    23.3690360

          ----------------------------------------------------------------------------



                        Rental Cost per SF, by NEWGRP & CAG, H-216 data                        8

                                                                  10:21 Monday, October 20, 1997



     Analysis Variable : RCSF





     NEWGRP  CAG       N Obs      N          Mean       Std Dev       Minimum       Maximum

     --------------------------------------------------------------------------------------

     CD1     A          1005   1005     8.9801650     7.2494492     0.0083333    35.7997936



             B           579    579     9.0047196     6.7200923     0.0076923    34.0069204



             C          1000   1000     9.3893597     6.8595495     0.0083333    35.2941176



             D           464    464     8.5229304     6.5887241     0.0104167    33.7500000



     CD2     E           778    778     7.8746352     5.8485123     0.0100000    32.9489292



             F           927    927     7.3395279     4.9658208     0.0076923    34.4827586



             G          2077   2077     6.6074240     3.6282007     0.1449275    33.1321429



     CD3     H          2901   2901     6.0872009     2.9960107     0.8760470    26.6166667



             J          3659   3659     5.7832173     2.6330187     0.8841733    19.8545259



             K          6115   6115     5.6662241     2.6889097     0.8640000    17.6470588



             L           715    715     5.6253164     2.9428718     1.2500000    17.8200000

     --------------------------------------------------------------------------------------

�                    Rental Cost per SF, by Delivery group & CAG, H-216 data                    9

                                                                  10:21 Monday, October 20, 1997



   Analysis Variable : RCSF





   DGROUP      CAG       N Obs      N          Mean       Std Dev       Minimum       Maximum

   ------------------------------------------------------------------------------------------

   City-Other  A          1005   1005     8.9801650     7.2494492     0.0083333    35.7997936



               B           576    576     9.0207591     6.7292408     0.0076923    34.0069204



               C           988    988     9.4127495     6.8934089     0.0083333    35.2941176



               D           448    448     8.5662650     6.6654979     0.0104167    33.7500000



               E           691    691     7.8046926     6.0706852     0.0100000    32.9489292



               F           659    659     7.1101255     5.3549054     0.0076923    34.4827586



               G           911    911     6.0113233     3.8221505     0.1449275    33.1321429



               H           470    470     5.2054203     2.8742918     0.8760470    26.6166667



               J           142    142     4.7674242     2.5158799     0.8841733    19.8545259



               K           144    144     4.4408896     2.0843988     0.8640000    11.6883117



               L            16     16     4.3691224     2.1974571     1.2500000     8.1428571



   Non-city    B             3      3     5.9251363     4.2503510     1.4803597     9.9498715



               C            12     12     7.4635935     2.3351708     4.5135406    12.5000000



               D            16     16     7.3095605     3.7888043     1.6512605    13.3088042



               E            87     87     8.4301565     3.6062111     1.7006803    17.6916667



               F           268    268     7.9036181     3.7955560     1.3238017    17.1298963



               G          1166   1166     7.0731597     3.3987457     1.2860483    17.8618682



               H          2431   2431     6.2576809     2.9896852     1.3282326    17.8722003



               J          3517   3517     5.8242302     2.6297512     1.2850985    17.7976190



               K          5971   5971     5.6957749     2.6950834     1.2864078    17.6470588



               L           699    699     5.6540705     2.9526673     1.2847966    17.8200000

   ------------------------------------------------------------------------------------------

�                       Rental Cost per SF, by Delivery group, H-216 data                      10

                                                                  10:21 Monday, October 20, 1997



        Analysis Variable : RCSF





        DGROUP      N Obs      N          Mean       Std Dev       Minimum       Maximum

        --------------------------------------------------------------------------------

        City-Other   6050   6050     7.7266699     6.0742204     0.0076923    35.7997936



        Non-city    14170  14170     5.9972545     2.8858878     1.2847966    17.8722003

        --------------------------------------------------------------------------------

